

Multi-frequency C-V Measurements of Semiconductors

Application Note 369-5

**HP 4284A
Precision LCR Meter**

Introduction

Parameters such as the capacitance of the oxide layer (C_{ox}) and the density of substrate impurities (N_{sub}) that are required in the evaluation of the manufacturing process of MOS type semiconductors can be derived by using measured C-V characteristics. To make an accurate evaluation of these processes, precise C-V measurements are required. Such measurements entail the following difficulties.

Difficulties

1. There is no single instrument that can make C-V measurements from low to high frequencies.
2. It is difficult to compensate for the additional errors that occur when cable extensions or a probe are used.
3. The accuracy and reliability of the DC bias voltage are not sufficient to perform repeatable C-V measurements.

In the following discussion we will show how C-V measurements are performed with the HP 4284A solving these problems.

Solutions Offered by HP 4284A

1. Wide Frequency Range Measurements from 20 Hz - 1 MHz.

The program listing shown in appendix was used to measure the C-V characteristics at 10 kHz, 100 kHz and 1 MHz of the MOS

device whose characteristics are shown in Figure 1. Thus the HP 4284A can single handedly perform C-V measurements in both the low and high frequency ranges. This allows it to measure high-loss devices (semiconductors on large diameter wafers, etc.), which are difficult to measure at 1 MHz, at low frequencies (10 kHz, 100 kHz, etc.).




Figure 1. Multi-Frequency C-V Characteristics of a Semiconductor

2. Compensation Functions for Cable Extensions and Probers.

In order to test semiconductor devices on a wafer, an extension cable and a prober are required. (See Figure 2) The extension cable and the prober cause additional errors that greatly influence the test value. The HP 4284A's 2m/4m Cable Length Operation function (opt.006 with HP 16048D/E Test Leads) minimizes additional errors caused by using 2m/4m extension cables. The HP 4284A's OPEN/SHORT/LOAD compensation functions minimize errors caused by a prober. This powerful compensation function ensures highly accurate C-V measurements even when a prober is used.

3. Highly Accurate Internal Bias

Insufficient accuracy and stability of the bias voltage applied to a device prevent accurate C-V measurements. The HP 4284A ensures the application of a stable bias voltage with a maximum accuracy of 0.1% + 1 mV. (Opt.001) This minimizes measurement errors due to bias voltage errors.




Figure2. Extension Cable and Prober

Table 1. HP 4284A Opt.001 DC Bias Capability

Range	Resolution	Accuracy*
±(0.000 to 4.000)V	1mV	±(0.1% + 1mV)
±(4.002 to 8.000)V	2mV	±(0.1% + 2mV)
±(8.005 to 20.000)V	5mV	±(0.1% + 5mV)
±(20.01 to 40.000)V	10mV	±(0.1% + 10mV)

*This can only be used when the test signal level is < 2 Vrms.

Conclusion

The HP 4284A features a wide frequency range, powerful compensation functions and highly accurate bias characteristics. This versatility of the meter guarantees highly accurate C-V measurements, and higher efficiency in the evaluation of semiconductor processes. Since the HP 4284 can perform all of the low and high frequency C-V measurements singlehandedly, it will greatly contribute to the lowering of capital costs.

Appendix. Sample Program

```

10 ! ***** HP 4284A C-V MEASUREMENT SAMPLE PROGRAM *****
20 !
30 GCLEAR
40 PRINT CHR$(12) ! CLEAR DISPLAY
50 DIM C(3,128),Cn(3,128),Vbias(128)
60 REAL Datab(1:513) BUFFER
70 DIM Dum$(6) BUFFER
80 ASSIGN @Ad TO 717;FORMAT ON ! DEFINE I/O PATH
90 ASSIGN @Pat TO 717;FORMAT OFF ! DEFINE I/O PATH
100 ASSIGN @Dummy TO BUFFER Dum$ ! DEFINE BUFFER FOR HEADER
110 ASSIGN @Buff TO BUFFER Datab(*) ! DEFINE BUFFER FOR DATA
120 REMOTE @Ad ! SET HP4284A TO REMOTE MODE
130 OUTPUT @Ad;"*RST" ! RESET HP4284A
140 OUTPUT @Ad;"VOLT:LEV 0.3V" ! SET OSC LEVEL TO 300mV
150 OUTPUT @Ad;"OUTP:HPOW ON" ! BIAS OPTION ON
160 OUTPUT @Ad;"TRIG:SOUR BUS" ! HP-IB TRIGGER MODE
170 OUTPUT @Ad;"FORM:DATA REAL" ! BINARY DATA TRANSFER
180 OUTPUT @Ad;"MEM:DIM DBUF,101" ! DEFINE BUFFER IN HP4284A
190 OUTPUT @Ad;"MEM:FILL DBUF" ! ENABLE BUFFER TRANSFER
200 OUTPUT @Ad;"APER SHOR" ! SET INTEG TIME TO SHORT
210 GOSUB Corr
220 PRINT "CONNECT DUT AND PRESS CONTINUE"
230 PAUSE
240 PRINT CHR$(12) ! CLEAR DISPLAY
250 Holdtime=10 ! HOLD TIME = 10sec
260 Delaytime=.1 ! DELAY TIME = 100msec
270 Vbias(1)=-5 ! START BIAS VOLTAGE
280 Vstep=.1 ! 100mV BIAS STEP
290 OUTPUT @Ad;"TRIG:DEL "&VAL$(Delaytime) ! SET DELAY TIME
300 OUTPUT @Ad;"BIAS:STATE ON" ! BIAS ON
310 Freq$(1)=" 1MHz"
320 Freq$(2)=" 100KHz"
330 Freq$(3)=" 10KHz"
340 FOR I=1 TO 3
350 OUTPUT @Ad;"FREQ"&Freq$(I) ! FREQUENCY SETTING
360 OUTPUT @Ad;"BIAS:VOLT "&VAL$(Vbias(1)) ! SET START BIAS VOLTAGE
370 PRINT TABXY(5,15);"HOLD TIME 10SEC" ! HOLD TIME
380 WAIT Holdtime
390 PRINT TABXY(5,15);"MEASURING at "&Freq$(I)
400 FOR J=1 TO 101
410 OUTPUT @Ad;"TRIG" ! TRIGGER HP4284A
420 IF J=101 THEN 450
430 Vbias(J+1)=Vbias(J)+Vstep ! CHANGE BIAS VOLTAGE
440 OUTPUT @Ad;"BIAS:VOLT "&VAL$(Vbias(J+1)) ! SET BIAS VOLTAGE
450 NEXT J
460 RESET @Dummy
470 RESET @Buff
480 OUTPUT @Ad;"MEM:READ? DBUF" ! READ BUFFER
490 TRANSFER @Pat TO @Dummy;COUNT 6,WAIT ! TRANSFER HEADER
500 TRANSFER @Pat TO @Buff;COUNT 101*4*8+1,WAIT ! TRANSFER DATA
510 FOR J=1 TO 101
520 C(I,J)=Datab(J*4-3)
530 NEXT J
540 IF MAX(C(*) )=0 THEN Err ! CHECK IF MAX C VALUE IS 0
550 FOR J=1 TO 101
560 Cn(I,J)=C(I,J)/MAX(C(*) ) ! NORMALIZE C VALUE
570 ! PRINT Cn(I,J) ! PRINT NORMALIZED C VALUE
580 NEXT J
590 OUTPUT @Ad;"MEM:CLE DBUF;FILL DBUF" ! CLEAR DATA IN BUFFER
600 PRINT CHR$(12) ! CLEAR DISPLAY
610 NEXT I
620 OUTPUT @Ad;"BIAS:STATE OFF" ! BIAS OFF
630 GOSUB Plotting
640 GOTO Ending
650 !
660 ! ***** PLOT C-V *****
670 ! THE FOLLOWINGS ARE FOR PLOTTING THE MEASUREMENT DATA TO CRT.
680 !
690 Plotting:DEG ! SET ANGLE UNIT TO DEGREE
700 GRAPHICS ON
710 VIEWPORT 0,150,17,100 ! PLOT NUMBERS AND LABELS
720 WINDOW 0,100,0,100 !
730 LDIR 0 !
740 LOR 2 !
750 CSIZE 4 !
760 MOVE 50,15 !
770 LABEL "BIAS (V)" !
780 MOVE 8,23 !
790 LABEL "-5.0" !
800 MOVE 54,23 !
810 LABEL "0" !
820 MOVE 94,23 !
830 LABEL "5.0" !
840 MOVE 5,28 !

```

```

850 LABEL "0.0"
860 MOVE 5,98
870 LABEL "1.0"
880 MOVE 75,5
890 LABEL " 1MHz"
900 MOVE 75,10
910 LABEL "100kHz"
920 MOVE 75,15
930 LABEL " 10kHz"
940 MOVE 88,5
950 DRAW 98,5
960 LINE TYPE 8
970 MOVE 88,10
980 DRAW 98,10
990 LINE TYPE 3
1000 MOVE 88,15
1010 DRAW 98,15
1020 LINE TYPE 1
1030 LDIR 90
1040 MOVE 3,55
1050 LABEL "Cap/Cox"
1060 LDIR 0
1070 MOVE 5,28
1080 LABEL "0.0"
1090 MOVE 5,98
1100 LABEL "1.0"
1110 |
1120 VIEWPORT 23,125,40,100
1130 FRAME
1140 WINDOW -5,5,0,1
1150 AXES 1,.2,-5,0
1160 |
1170 FOR I=1 TO 3
1180 MOVE Vbias(I),Cn(I,1)
1190 FOR J=2 TO 101
1200 DRAW Vbias(J),Cn(I,J)
1210 NEXT J
1220 LINE TYPE 3
1230 IF I=1 THEN LINE TYPE 8
1240 NEXT I
1250 LINE TYPE 1
1260 RETURN
1270 |
1280 Corr:| ***** SUBROUTINE FOR CORRECTION *****
1290 | THE FOLLOWINGS ARE FOR PERFORMING OPEN/SHORT COMPENSATION.
1300 |
1310 OUTPUT @Ad;"DISP:PAGE CSETUP"
1320 OUTPUT @Ad;"CORR:LENG 0;METH SING"
1330 OUTPUT @Ad;"CORR:SPOT1:FREQ 1MAHz;STAT ON"
1340 OUTPUT @Ad;"CORR:SPOT2:FREQ 100KHz;STAT ON"
1350 OUTPUT @Ad;"CORR:SPOT3:FREQ 10KHz;STAT ON"
1360 OUTPUT @Ad;"CORR:LOAD:STAT OFF"
1370 PRINT "OPEN COMPENSATION"
1380 PRINT TABXY(5,15);"PRESS CONTINUE"
1390 PAUSE
1400 PRINT TABXY(5,15);" "
1410 OUTPUT @Ad;"CORR:SPOT1:OPEN"
1420 OUTPUT @Ad;"CORR:SPOT2:OPEN"
1430 OUTPUT @Ad;"CORR:SPOT3:OPEN"
1440 OUTPUT @Ad;"*OPC?"
1450 ENTER @Ad;AS
1460 OUTPUT @Ad;"CORR:OPEN:STAT ON"
1470 PRINT CHR$(12)
1480 PRINT "SHORT COMPENSATION"
1490 PRINT TABXY(5,15);"PRESS CONTINUE"
1500 PAUSE
1510 PRINT TABXY(5,15);" "
1520 OUTPUT @Ad;"CORR:SPOT1:SHOR"
1530 OUTPUT @Ad;"CORR:SPOT2:SHOR"
1540 OUTPUT @Ad;"CORR:SPOT3:SHOR"
1550 OUTPUT @Ad;"*OPC?"
1560 ENTER @Ad;AS
1570 OUTPUT @Ad;"CORR:SHOR:STAT ON"
1580 OUTPUT @Ad;"DISP:PAGE MEAS"
1590 PRINT CHR$(12)
1600 RETURN
1610 |
1620 | *****
1630 Err:PRINT "C-V MEASUREMENT WAS FAILED."
1640 Ending:END

```

For more information about Hewlett-Packard Test and Measurement products, applications, services, and for a current sales office listing, visit our web site, <http://www.hp.com/go/tmdir>. You can also contact one of the following centers and ask for a test and measurement sales representative.

United States:
Hewlett-Packard Company
Test and Measurement Call Center
P.O. Box 4026
Englewood, CO 80155-4026
1 800 452 4844

Canada:
Hewlett-Packard Canada Ltd.
5150 Spectrum Way
Mississauga, Ontario
L4 W 5G1
(905) 206 4725

Europe:
Hewlett-Packard
European Marketing Center
P.O. Box 999
1180 AZ Amstelveen
The Netherlands
(31 20) 547 9900

Japan:
Hewlett-Packard Japan Ltd.
Measurement Assistance Center
9-1, Takakura-Cho, Hachioji-Shi,
Tokyo 192, Japan
Tel: (81-426) 56-7832
Fax: (81-426) 56-7840

Latin America:
Hewlett-Packard
Latin American Region Headquarters
5200 Blue Lagoon Drive
9th Floor
Miami, Florida 33126
U.S.A.
(305) 267 4245/4220

Australia/New Zealand:
Hewlett-Packard Australia Ltd.
31-41 Joseph Street
Blackburn, Victoria 3130
Australia
1 800 629 485

Asia Pacific:
Hewlett-Packard Asia Pacific Ltd
17-21/F Shell Tower, Times Square,
1 Matheson Street, Causeway Bay,
Hong Kong
Tel: (852) 2599 7777
Fax: (852) 2506 9285

Data subject to change
Copyright © 1998
Hewlett-Packard Company
Printed in U.S.A. 6/98
5950-2953